1 Sieve of Eratosthenes

The following code generates a list of primes up to n using the Sieve of Eratosthenes. A set of integers up to n is constructed in the variable P. We then remove higher multiples of the smallest element of P. Here we used the minus operation for sets to eliminate the composite numbers.

A list called primes is used to store the primes as they are generated. At each step, the smallest number, m, in the sieved set P is a prime. We add this to the list of primes. Then we remove all the multiples of m from P. Then the smallest number in P must be a prime. This process is repeated until the smallest number exceeds the square root of n. Then the remaining numbers in P must all be prime. Hence we add this to the list of primes and return the combined list. The returned list may not be sorted, but it can be arranged in increasing order using the sort command.

```
> sieveoferatosthenes := proc(n)
> local t, P, i, x, k,m, primes;
>  t:= floor(sqrt(n));
> P:= {seq(i, i=2..n)};
> primes:=[ ];
> while min( op(P)) <= t do m:= min(op(P)); primes:=[op(primes),m];
>  x:= {seq( k* m , k=1..floor(n/m) ) };
> P:= P minus x;
>
> od;
> RETURN( [op(primes), op(P)]);
> end;
```

```
sieveoferatosthenes := proc(n)
                       local t,P,i,x,k,m,primes;
                           t := floor(sqrt(n));
                           P := {seq(i,i = 2 .. n)};
                           primes := [];
                           while min(op(P)) <= t do
                               m := min(op(P));
                               primes := [op(primes),m];
                               x := {seq(k*m,k = 1 .. floor(n/m))};
                               P := P minus x
                           od;
                           RETURN([op(primes),op(P)])
                       end
```

```
> sieveoferatosthenes(10);
```

$$[2, 3, 7, 5]$$

```
> sieveoferatosthenes(100);
```

$$[2, 3, 5, 7, 41, 31, 43, 17, 19, 29, 37, 11, 13, 59, 61, 47, 23, 53, 67, 71, 73, 79, 83, 89, 97]$$

We can arrange the above list in increasing order using the sort command.

```
> sort(");
```

$$[2, 3, 5, 7, 11, 13, 17, 19, 23, 29, 31, 37, 41, 43, 47, 53, 59, 61, 67, 71, 73, 79, 83, 89, 97]$$

We also used the min function, the op function, and the minus operation on sets.